
Iskolai focibajnokság 

1.  A foci.txt  állományt megnyitásakor konvertálási varázslót alkalmazunk (elválasztójel a 
tabulátor-karakter). Beolvassuk az adatokat, elmentjük bajnoksag  néven a használt 
programnak megfelelıen. 

2.  A C4:C6-os tartományba beírjuk az értékeket. A C8-as cellába beírhatjuk a csapatok számát, de 
a jövıbeli használhatóság szempontjából jobb, ha függvényt használunk: 
C8: =DARAB2(B11:B22) 

3.  I11: =G11-H11  

4.  J11: =D11*3+E11  

5.  Ahhoz, hogy a képlet csapatszám-független legyen, abszolút hivatkozást kell használnunk. 
K11: =C$8-1-C11  

6.  Ahhoz, hogy a képlet más büntetıpont-rendszer esetén is használható legyen, abszolút 
hivatkozást kell használnunk. 
N11: =K11*C$6+L11*C$5+M11*C$4  

7.  O11: =J11-N11  

8.  P11: =(C$8-1)*3-O11  

9.  Q11: =J11/((C$8-1)*3) 

A kapott értékeket százalék formátumban jelenítjük meg, tizedesjegyek nélkül. 

10.  Az összpontszám alapján csökkenı sorrendbe rendezzük az adatokat. 

11.  A használt táblázatkezelı programnak megfelelıen elvégezzük a kijelölt formázásokat. 
 

Vízállás 

1.  A szövegállomány megnyitásakor konvertálási varázslót alkalmazunk. A cellák elválasztó jele a 
tabulátor-karakter. Az importálás után, az állomás mentésekor ügyeljünk a megfelelı típus 
kiválasztására! 

2.  A hónapnevek és a napsorszámok beszúrásához 1 sort és 1 oszlopot szúrunk be. 

3.  A sorozat elsı két értékét beszúrva kitöltést alkalmazhatunk. Az írás irányát a cellák 
formázásánál állíthatjuk. 

4.  A sorozat elsı két értékét beszúrva, kitöltést alkalmazhatunk. 

5.  A Szerkesztés/Csere eljárással a rövidítést üres tartalomra cseréljük. 

6.  B34: =ÁTLAG(B2:B32) 
B35: =MIN(B2:B32) 
B36: =MAX(B2:B32) 

A B34 cella formátumát 1 tizedesjegyes megjelenítésre állítjuk. A 3 cella másolásával, a relatív 
hivatkozás következtében helyes eredményt kapunk. 

7.  Fontos, hogy az átlagok átlaga elvileg nem ad helyes eredményt, hiszen a hónapok eltérı számú 
napot tartalmaznak. 
Például: 
G38: =ÁTLAG(B2:M32) G39: =MIN(B2:M32) G40: =MAX(B2: M32) 

8.  A diagramokat a mintának megfelelı típussal készítsük el (vonal- illetve oszlopdiagram)! A 
varázsló indítása elıtt érdemes a megfelelı adattartományokat kijelölni. 
Ügyeljünk a tengelyfeliratozásra, diagramcímre és most jelmagyarázatot sem kérünk. 

9.  A szegélyezést az eszközsor ikonjával vagy a Formátum/Cellák/Szegély almenü paramétereivel 
végezhetjük. 

10.  Az oszlophatárolók mozgatásával, vagy a Formátum/Oszlop almenü használatával oldhatjuk 
meg a feladatot. 

11.  A Nézet/Élıfej és élıláb… almenü Egyéni élıfej funkcióját alkalmazzuk. A dátum beszúrását a 
Dátum objektum beillesztésével végezzük el. 

Web-szerver 

1.  A szövegállomány megnyitásakor konvertálási varázslót alkalmazunk. Az importálás után, az 
állomány mentésekor ügyeljünk a megfelelı típus kiválasztására! 

2.  Az óráknak és a forgalmi adatok százalékának 1-1 oszlopot és a címeknek 1 sort szúrunk be. 

3.  A címek 2-2 oszlophoz tartoznak, ezért cellaösszevonást alkalmazunk. 

4.  A sorozat elsı két értékét beszúrjuk, és kitöltést alkalmazunk. 

5.  A nagy számok olvashatóságát az ezres csoportosítással segítjük elı. 

6.  B33: =SZUM(B2:B32) 

A másik 3 cellába (D33, F33, H33) másolva a relatív hivatkozás eredményeként helyes 
eredményt kapunk. 

7.  Ha a nevezıben az összeg cellájára a sor rögzítésével hivatkozunk, akkor nemcsak oszlopon 
belül másolható a kifejezés, hanem a többi oszlopba is. 
Például: 
C2: =B2/B$33 

8.  A szegélyezés többszöri tartománykijelölést igényel. 

9.  A varázsló indítása elıtt érdemes a megfelelı adattartományokat kijelölni. A diagramokat a 
mintának megfelelı típussal készítsük el. 
Ügyeljünk, hogy a tengely függılegesen ne legyen feliratozva és most jelmagyarázatot sem 
kérünk. A diagramokról a napi letöltési adatmennyiségek változásának tendenciáit figyelhetjük 
meg. 

10.  A címek háttérszínének állítása után a diagram oszlopainak színét is változtassuk meg! 

 
Fagyi-Gofri 

1.  Egy új táblázatot kezdünk, a megfelelı néven és formátumban mentünk. 

2.  A kitöltés funkcióját felhasználva készítsük el a keretet. 

3.  A megadott cellákba írjuk be az árakat illetve a feliratokat! 

4.  A képlet elkészítésénél ügyeljünk az abszolút és a vegyes hivatkozások megfelelı kezelésére. 
Például a C3-as cellába az =$A$2*$B3+$B$1*C$2  képletet írva hibamentesen másolható képletet 
kapunk. 

5.  Állítsunk be az A2, B1 valamint a C3:M13 tartomány celláira pénznem formátumot, ezres 
tagolással és nulla tizedesjeggyel! 
A szövegek formázásánál alkalmazzunk cellaegyesítést! A „GOFRI” szónál állítsuk be az 
írásirányt fentrıl lefelé! A számokat igazítsuk függılegesen és vízszintesen középre! 
A szövegeket, az adagokat és az egységárakat emeljük ki a betők méretének növelésével és 
félkövér stílussal! 
Az adagokat tartalmazó cellák mintázatát állítsuk világosszürkére! 
A keretezésnél a külsı keret legyen vastag, belül vékony vonal! Az adagokat dupla vonal 
válassza el a számított értékektıl! 

Bumm! 

1.  A játékot a feladat megoldása elıtt érdemes kipróbálni. Nehezebb változata, ha a tiltott számot, 
mint számjegyet tartalmazó értékeknél is Bumm!-ot kell mondani. 
Ehhez a feladathoz nincs forrásállomány. Az állomány mentési útvonalát és nevét a feladat 
megoldásának elején állítsuk be! 

2.  Ahhoz, hogy a játék szimulációja megfelelıen frissüljön, az A2 és B2 cellákra kell majd 
hivatkoznunk. 

3.  D1: =A2 
D2: =D1+1 

A D2 cella tartalmát másoljuk a tartomány további celláiba. 

4.  E1: =HA(MARADÉK(D1;B$2)=0;"Bumm!";D1) 

Az oszthatóságot a =MARADÉK() függvénnyel vizsgáljuk. A másolhatóság miatt a tiltott 
számra abszolút vagy vegyes módon kell hivatkozni. 

5.  A szegélyezést az eszközsor ikonjával vagy a Formátum/Cellák…/Szegély almenü 
paramétereivel végezhetjük. 
A „Bumm!” felirat színét feltételes formázással tudjuk változtatni: 

 

6.  Az oszlophatárolók mozgatásával, vagy a Formátum/Oszlop almenü használatával oldhatjuk 
meg a feladatot. 

 
Piac/1 

1.  Az ar.txt  állományt megnyitjuk, konvertáljuk, beolvassuk az adatokat, elmentjük piac1  
néven a használt programnak megfelelıen. 

2.  A rovatcímeket kijelöljük, írásirányát 90o-kal balra forgatjuk. Az adatokat a cellahatárok között 
is középre helyezzük. 

3.  I4: =KEREK(ÁTLAG(B4:H4);0) 

Ha a kerekítést nem függvénnyel, hanem a számformátum beállításával oldjuk meg, akkor csak 
látszólagosan kerekít a táblázatkezelı: a tizedesjegyeket továbbra is tárolja, csak nem jeleníti 
meg. A késıbbi számolásoknál ez hibás eredményhez vezethet! 
J4: =MAX(B4:H4) 
K4: =MIN(B4:H4) 

4.  B25: =KEREK(ÁTLAG(B4:B23);0) 
I25: =KEREK(ÁTLAG(B4:H23);0) 

A heti átlag az összes termék összes napi árának az átlaga, ami nem feltétlenül egyezik meg a 
napi átlagárak átlagával! Az eltérés oka az, hogy vannak olyan termékek, amelyek bizonyos 
napokon nem voltak kaphatók. Ezért az átlagot mindenképpen az összes adat 
figyelembevételével számoljuk! 

5.  B26: =HA(B25>$I25;"drága";HA(B25<$I25;"olcsó";"átla gos")) 

6.  L4: =DARABÜRES(B4:H4) 
B27: =DARAB(B4:B23)  

7.  M4: =J4-I4 
N4: =M4/I4 
O4: =I4-K4 
P4: =O4/I4 

Az N4-es és a P4-es cellára állítsunk be %-os megjelenítést, tizedesjegyek nélkül. 

8.  Q4: =N4+P4  

9.  B30: =DARABTELI(L4:L23;0) 
B32: =DARABTELI(Q4:Q23;">0") 

10.  Az alábbi formázásokat kell végrehajtanunk: 
• a címet helyezzük az oszlopok között középre, vastagítsuk meg és mérete nagyobb legyen, 

mint az adatoké; 
• a termékek nevét vastagítsuk meg, ezeket és a rovatcímek hátterét színezzük 

halványszürkére; 
• az eredetileg megadott árakat vastagítsuk és dıltsük meg; 
• azokban a cellákban, amelyekben pénz jellegő adat szerepel, állítsunk be pénznem 

formátumot (tizedesjegyek nélkül); 
• az oszlopszélességet úgy állítsuk be, hogy a megadott árak és a számolt értékek mind 

látszódjanak; 
• keretezzük a táblázatot egyszeres vastagságú vonallal! 

 

Születésnap 

1.  Ehhez a feladathoz nincs forrásállomány. Egy új táblázatot kezdünk, a megadott néven elmentjük 
a munkafüzetet. 

2.  Az adatokat beírjuk a megfelelı cellákba. 
Az A18:A24-es tartomány elsı két cellájának értékét beírva kitöltést alkalmazhatunk, vagy az 
A18 cella ártékének beírása után az A19-es cellába kerülı képlet: =A18+1. 

Ha a táblázatkezelı program engedi, a hét napjait automatikus kitöltéssel a legkönnyebb bevinni, 
így elég a legelsı napot begépelni (hétfı). Ha nem ismer ilyen funkciót a program, akkor 
gépeljük be a hét napjait. 

3.  Félkövér valamint dılt betőtípust használunk. Az oszlopszélességnél a „Legszélesebb kijelölt” 
(„Optimális szélesség”) opciót választjuk. 

4.  B1: =MA() 

A B2-es cellába a születési dátum kerüljön, ügyeljünk a táblázatkezelı által használt dátum 
formátum elválasztójelére (általában pont). 

5.  C5: =B1-B2 

Az eredményt dátum formátumban kapjuk, ezt állítsuk át szám formátumúra, tizedesjegyek 
nélkül. 

6.  C6: =KEREK.LE(C5/7;0) 

vagy 
C6: =INT(C5/7) 

7.  C7: =ÉV(B1-B2)-1900 

Az eredményt dátum formátumban kapjuk, ezt állítsuk át szám formátumúra, tizedesjegyek 
nélkül. 

8.  A C5, C6, C7-es cellákban Egyéni (Felhasználói) számformátumot használunk. Például a C5-ös 
cellához használt formátumkód: 
# ##0" nap" vagy # ##0\ "nap" 

9.  C10: =HÉT.NAPJA(B2;2) 
C11: =FKERES(C10;A18:B24;2) 

A hét napjának nevét valójában a segédtáblázat felhasználása nélkül is meghatározhatjuk. Ehhez 
a C11-es cella értékét állítsuk a születésnapunk dátumára (azaz a képlet: =B2), és erre a cellára 
alkalmazzunk Egyéni (Felhasználói) formátumot nnnn formátumkóddal! 

10.  C14: =DÁTUM(ÉV(B2)+C7+1;HÓNAP(B2);NAP(B2)) 

11.  C15: =C14-B1 

Az eredményt dátum formátumban kapjuk. 
Ezután állítsunk be Egyéni (Felhasználói) számformátumot, a formátumkód: 
# ##0" nap" vagy # ##0\ "nap"  

Lakosok 

1.  A szövegállomány megnyitásakor konvertálási varázslót alkalmazunk. A cellák elválasztójele a 
tabulátor karakter. Az importálás után az állomány mentésekor ügyeljünk a megfelelı típus 
kiválasztására! 

2. A D oszlopot hagyjuk üresen, ez választja el a forrásadatokat a feladatok megoldásaitól. 
Minden választ a cellája elıtt vagy fölött magyarázószöveggel lássunk el! 

3. =SZUM(B2:B142) 

4. =AB.DARAB2(A1:C142;B1;H3:H4) , ahol a kritérium tartomány 
H3: Lakónépesség 
H4: >50000 

vagy 
=DARABTELI(B2:B142;">50000")  

5.  Vízszintesen felsoroljuk a hét régió nevét és a fölöttük levı cellákba a „Régió” címet írjuk. Ezzel 
hat kritériumtartományt hoztunk létre. Pl.: 
F6: Régió 
F7: Észak-Magyarország 
F8: =AB.MAX($A$1:$C$142;$B$1;F6:F7) 
F9: =INDEX($A$1:$C$142;HOL.VAN(F8;$B$1:$B$142;0);1)  

Az F8:F9 cellákat másoljuk a többi kritériumtartomány alá. Az abszolút hivatkozások a 
másolhatóság miatt szükségesek. 

6.  A kritériumtartományok felhasználásával a 
F10: =AB.SZUM($A$1:$C$142;$B$1;F6:F7) 

cellát másoljuk a többi régió neve alá. 
vagy 
=SZUMHA($C2:$C142;F7;$B2:$B142)  

7.  Irányított szőrés, majd rendezés a lakónépesség szerint csökkenıen. 

8.  F29: ide írjuk a keresett település nevét 
F30: =FKERES(F29;A1:C142;3;HAMIS) 

Az FKERES() függvény utolsó paraméterét HAMIS értékre kell állítani, hogy csak teljes 
névegyezés esetén kapjunk értéket. 

9.  A szegélyezés az adatok és a megoldások jó megfigyelhetısége miatt szükséges. A megoldás 
nem fér ki egy nyomtatási oldalra. 

 
TAJ szám 

1.  A személyi adatok nyilvántartásához használt azonosító jelek képzési szabályaihoz forrás a CD 
jogtár, 3. számú melléklet az 1996. évi XX. Törvényhez. 

2.  Ehhez a feladathoz nincs forrásállomány. Az állomány mentési útvonalát és nevét a feladat 
megoldásának elején állítsuk be! 

3.  Az Ellenırzés munkalapot a TAJ szám beviteléhez és az eredmény megjelenítéséhez használjuk. 

4.  Ha az A2 cellát szám formátumúnak hagynánk, akkor a bevezetı 0 számjegyeket a rendszer 
elhagyná. 

5.  A Feldolgozás munkalapon: 
B2: =ÉRTÉK(KÖZÉP(Ellen ırzés!$A$2;A2;1)) 

A számjegyeket a =KÖZÉP() függvénnyel állítjuk elı. A karakter számmá konvertálását az 
=ÉRTÉK() függvénnyel oldjuk meg. 
Ügyeljünk az abszolút hivatkozás használatára! 

6.  A törvény szövege szerint ez 3 és 7. 

7.  C2: =HA(MARADÉK(A2;2)=1;B2*$E$2;B2*$E$3) 

Az oszthatóságot a =MARADÉK() függvénnyel vizsgáljuk. A másolhatóság miatt az E2 és E3 
cellára abszolút módon kell hivatkozni. 
C11: =SZUM(C2:C9) 

8.  Az Ellenırzés munkalapon: 
B2: =HA(MARADÉK(Feldolgozás!C11;10)=Feldolgozás!B10 ;"Jó";"Téves") 

9.  Az Ellenırzés munkalapon érdemes nagyobb karakterméretet és félkövér stílust alkalmazni az 
A1:B2 cellákra. 
A Feldolgozás munkalapon az oszlopszélességeket úgy állítsuk be, hogy minden adat olvasható 
legyen! 

 
Tessék bet őzni! 

1.  A szövegállomány megnyitásakor konvertálási varázslót alkalmazunk, az elválasztójel a 
tabulátor karakter. Az importálás után, mentéskor ügyeljünk a megfelelı típus kiválasztására! A 
fájl az ábécé betőit és a hozzájuk rendelt szavak listáját tartalmazza. 

2.  A szövegbeolvasáshoz sort és a betőzéshez oszlopokat kell a névlista elé beszúrni. 

3.  A két cellát összevonjuk, hogy a hosszabb szöveg elférjen. 

4.  A sorozat elsı két értékét beszúrva kitöltést alkalmazhatunk. 

5.  B4: =KÖZÉP($C$1;A4;1) 

A B4 cellában szövegkezelı függvényt alkalmazunk. A másolás a C1 cella rögzítését tételezi fel. 
Az A oszlopban a bető sorszámára kell hivatkoznunk. 

6.  A problémát keresıfüggvény alkalmazásával oldjuk meg. 
Például: 
C4: =  FKERES(B4;$E$3:$F$31;2); 

Szóközt tartalmazó vagy üres cellákban ne jelenjen meg hibajelzés! 
Például: 
C4: =HA(ÉS(B4<>" ";B4<>"");FKERES(B4;$E$3:$F$31;2);"")  

7.  A szegélyezést az eszközsor ikonjával vagy a Formátum/Cellák/Szegély almenü paramétereivel 
végezhetjük. 
A nem üres cellák szegélyezéséhez feltételes formázást használunk: 

 
8.  Az oszlophatárolók mozgatásával, vagy a Formátum/Oszlop almenü használatával oldhatjuk 

meg a feladatot. 

Rikiki 

1.  A pontosvesszıvel tagolt forrásállományt beolvassuk. Mentés a használt programnak 
megfelelıen. 

2.  Beszúrunk egy üres oszlopot a táblázat elé. Négy játékosnál a maximálisan kiosztható lapok 
száma 13. Így az A oszlopban 1-tıl 13-ig, majd csökkenıen 1-ig szerepelnek a számok. A 
sorozatok elsı két tagját beírva kitöltést alkalmazhatunk, vagy pedig képlettel: az A3 cellába 1-et 
írunk, és: 
A4: =A3+1 
A16: =A15-1  

3.  Beszúrjuk az üres oszlopokat a játékosok adatai mögé (azaz üres oszlopok lesznek: D, G, J, M). 

4.  D3: =HA(B3=C3;10+2*B3;-2*ABS(B3-C3)) 

A képlet nemcsak egy játékosnál másolható az összes fordulóra, hanem a cellák helyzetébıl 
adódóan a többi játékoshoz is másolható (azaz a G, J, M oszlopokba is). 

5.  D29: =SZUM(D3:D27) 

6.  Elkészítjük a feliratokat az A31, A33, A35, A37 cellákban (pl.: Legmagasabb pontszám, A játék 
nyertese, Legalacsonyabb pontszám, A játék vesztese). 
B32: =MAX(D29;G29;J29;M29) 
B34: =INDEX(B1:M29;1;HOL.VAN(B32;B29:M29;0)-2) 

A játékosok nevei a B, E, H, K oszlopokban vannak, azaz 2 oszloppal elırébb („balrább”), mint a 
pontszám, ezért az oszlopindexbıl 2-t le kell vonni. 
B36: =MIN(D29;G29;J29;M29) 
B38: =INDEX(B1:M29;1;HOL.VAN(B36;B29:M29;0)-2)  

7.  Az alábbi formázásokat kell végrehajtanunk: 
• a játékosok neveit helyezzük az adataikat tartalmazó 3-3 oszlop között középre,  
• a 2. sortól kezdve az adatokat helyezzük vízszintesen középre, 
• az elsı két sor adatait vastagítsuk meg, 
• a mintának megfelelıen keretezzünk egyszeres és dupla vonallal! 

8.  Fekvı tájolást, vízszintes és függıleges középre igazítást állítunk be a lapra. Szükség szerint 
beállítjuk a nagyítás százalékos mértékét. 

9.  A kívánt adatokat elhelyezzük az élıfejben és az élılábban (Nézet/Élıfej és élıláb…) 
 

Hardver 

1.  A tabulátorral tagolt forrásállomány beolvasásakor konvertálási varázslót alkalmazunk. 

2.  Átnevezzük a munkalapot. 

3.  Egy új munkalapot szúrunk be és átnevezzük. 

4.  A tabulátorral tagolt szövegállományt konvertálási varázslóval megnyitjuk, és az adatokat a 
Megrendelés munkalapra másoljuk. Dolgozhatunk úgy is, hogy valamilyen 
szerkesztıprogrammal megnyitjuk a megrend.txt  állományt, majd a vágólapon keresztül 
másoljuk az adatokat a munkalapra. Mindenképpen ügyeljünk arra, hogy a beillesztés az A1-es 
cellától kezdıdıen történjen! 

5.  Elvégezzük a kijelölt formázásokat (Formátum/Cellák… menüpont). Az egész táblázatot 
kijelölve egyszeres szegéllyel keretezünk.  
A kategórianevek (Merevlemez, Alaplap stb.) celláit egyesítjük, az oszlopszélességet a 
legszélesebb adat szélességére állítjuk (Formátum/Oszlop/Legszélesebb kijelölt). 

6.  Elvégezzük a kijelölt cellaformázásokat. 

7.  A termék pontos megnevezését a Megrendelés munkalap A illetve B oszlopában lévı adatok 

együttesen határozzák meg. Ezeket a címkéket a Listák munkalapban kell megkeresnünk. Pl.: 
C3: =FKERES($B3;Listák!$A$3:$D$19;Megrendelés!C$1;I GAZ) 

Sajnos, az egyes kategóriáknál különbözı a keresési tartomány, így a C oszlop celláiban más-
más paraméterezéső függvényt kell szerepeltetnünk. A keresési tartományok pl. merevlemez 
esetén Listák!$A$3:$D$19, alaplap esetén Listák!$A$21:$D$36  stb. 
C4: =FKERES($B4;Listák!$A$21:$D$36;Megrendelés!C$1; IGAZ) 

Ezek a függvények viszont átmásolhatók az E illetve F oszlopokba, mivel a keresési 
függvényhez az oszlopszámot az elsı sor tartalmazza! Itt már csak a darabszámmal kell 
beszoroznunk a kapott értékeket, ami a D3 cellában található. 
E3: 
=D3*FKERES($B3;Listák!$A$3:$D$19;Megrendelés!E$1;IG AZ) 
F3: 
=D3*FKERES($B3;Listák!$A$3:$D$19;Megrendelés!F$1;IG AZ)  

8.  Az elsı sorba beírt oszlopszámok az elızı képletek megalkotását könnyítik meg, de esztétikai 
szempontból elınyösebb, ha nem látszódnak. Ezért betőszínüket fehérre állítjuk. 

9.  E20: =SZUM(E3:E19) 
F20: =SZUM(F3:F19) 

10.  Az E és F oszlop celláira pénznem formátumot állítunk be, ezres tagolással, tizedesjegyek 
nélkül. 

11.  Ügyeljünk, hogy a mentés a táblázatkezelı saját formátumában történjen! A fájl neve az egyik 
forrásállományéval azonos lesz, csak a kiterjesztése lesz más. 

 
Tızsdei árfolyamok 

1.  A pontosvesszıvel tagolt forrásállományt beolvassuk. 

2.  Végrehajtjuk a kijelölt karakterformázásokat, igazításokat. Az oszlopszélességet úgy állítjuk be, 
hogy minden oszlopban minden adat olvasható legyen. 
A kiugróan magas átlagárú napokat feltételes formázással emeljük ki: 

 
3.  A megfelelı májusi adatok kijelölése után (A84:A104; E84:E104, nem összefüggı tartomány!) 

diagramvarázslóval dolgozunk. Grafikon típust válasszunk, ne felejtkezzünk meg a címekrıl 
(diagramcím, tengelyfeliratok), valamint a szürke háttérrıl és a piros vonalszínrıl! 
Jelmagyarázatra nincs szükség. A diagram méretét úgy válasszuk meg, hogy a vízszintes 
tengelyen a dátumok is jól olvashatók legyenek! 

4.  A Diagram/Trendvonal felvétele… menüpont segítségével lineáris trendvonalat veszünk fel. 

5.  Munkánkat a megadott néven mentjük. Ügyeljünk arra, hogy a mentés a táblázatkezelı saját 
formátumában történjen! 

Sportlövészet 

1.  A tabulátorral tagolt forrásállomány beolvasásakor konvertálási varázslót alkalmazunk. 

2.  Az 1. és a 12. sor, valamint az F, H oszlopok adatait megvastagítjuk, középre helyezzük, 
sortörést állítunk be. A G oszlop adatai dılt betősek. Az A1 és A12 cellákban két eltérı, jól 
megkülönböztethetı betőszínt használunk. A D2:F11, H2:H6, D13:F22, H13:H17 tartományok, 
valamint a C24 cella háttérszínét sárgára állítjuk.  
Az F illetve H oszlop szám jellegő adatokat tartalmazó celláira egyéni számformátumot állítunk 
be, formátumkód pl.: #0,0\ "kör" 

3.  D2: =10-GYÖK(B2*B2+C2*C2) 
E2: =KEREK(D2;1) 
F2: =HA(E2>=0;E2;0) 
H2: =MAX(F2:F11) 
H3: =NAGY(F2:F11;2) 
H4: =MIN(F2:F11) 
H5: =ÁTLAG(F2:F11) 
H6: =SZUM(F2:F11) 
C24: 
=HA(H6=H17;”új sorozat”;HA(H6>H17;A1;A12))  

4.  Pont(XY)/XY Diagram típusú, összekötés nélküli diagramot rajzolunk. Az adatsorokat a 
következıképpen állítjuk be: 
1. adatsor:  
Név: =szetlov!$A$1 
X értékek: =szetlov!$B$2:$B$11 
Y értékek: =szetlov!$C$2:$C$11 

2. adatsor: 
Név: =szetlov!$A$12 
X értékek: =szetlov!$B$13:$B$22 
Y értékek: =szetlov!$C$13:$C$22 

A diagram méretét úgy válasszuk meg, hogy vízszintes és függıleges irányban az egység azonos 
legyen! Így a lövések elhelyezkedése arányos a céltáblán látottakkal. 
Az adatok között szerepel több olyan lövés is, amely nagyon közel helyezkedik el a céltábla 
közepéhez (Grigorjeva 10. lövése, Hesse 6. és 10. lövése). Ezek a diagram kis mérete miatt 
nehezen (vagy egyáltalán nem) különböztethetık meg. 

5.  A diagram létrehozása után az egyes adatsorok színét a versenyzık nevének betőszínével 
megegyezıen állítjuk be (Adatsorok formázása/Mintázat/Jelölı). 

6.  A megadott néven és formátumban mentünk. 

Keresd a hibát! 

 A helytelen, illetve a helyes formázás  

1. 

 
 

 

A diagram áttekinthetetlen, zsúfolt. Módosítási 
javaslatok: 

• A jelmagyarázat rossz helyen van. Sokkal több 
hely lenne az adatoknak, ha pl. lent helyeznénk 
el. 

• Az x-tengely feliratai nem látszódnak a kevés 
hely miatt. Ha a jelmagyarázat áthelyezése nem 
oldja meg teljesen a problémát, akkor 
használjunk kisebb betőméretet! 

• Az y-tengelyskálán a tizedesjegyeket fölösleges 
szerepeltetni, ezzel is a diagram területe nı. 

• A túl sok rácsvonal áttekinthetetlenné teszi a 
diagramot. A vízszintes segédrácsok, és a 
függıleges rácsvonalak fölöslegesek. 

• Zavaró a háttér. 
• A feliratok nem látszódnak. Húzzuk szét ıket 

(pl. az egyik adatsor feliratait lejjebb helyezzük 
el), ekkor viszont az oszlopok átszínezése is 
szükséges lehet (túl sötétek). Az olvashatóságot 
a kis betőméret is rontja, ezen viszont nem 
tudunk segíteni a sok adat miatt. Valójában az 
ilyen jellegő diagramoknál nem érdemes 
feliratozni az oszlopokat! 

 

2. 

 
 

 

Ezekhez az adatokhoz nem megfelelı diagramtípust 
választottunk! A kördiagramot akkor használjuk, ha 
azt szemléltetjük, hogy milyen arányban „veszik ki 
részüket” az egészbıl az egyes részadatok (pl. az 
összes hiányzásból az egyes osztályok). 
A tanulmányi átlagok egymástól független adatok, 
így nincs értelme, hogy az összeghez való 
viszonyukat ábrázoljuk.  
A három átlag majdnem azonos körcikk-területet 
eredményez, így a diagram semmit sem szemléltet.  
Szerencsésebb volna más típust használni, pl. oszlop- 
vagy vonaldiagramot. Itt már jól látszik az átlagok 
közötti különbség. 

3. 

 
 

 

A diagram láthatóságával nincs probléma, csak egy 
kicsit megtévesztı. Elsı ránézésre azt 
gondolhatnánk, hogy a három osztály tanulmányi 
eredménye között óriási a különbség. Az adatokat 
nézve azonban látható, hogy az átlagok közötti 
eltérés 0,1-nél is kevesebb. (Ilyen jellegő 
megtévesztı diagramokkal gyakran találkozunk 
újságokban, televízióban, interneten. Ne higgyünk 
mindig az elsı benyomásnak!) 
Az „átverést” az okozza, hogy az y-tengelyskála 
kezdıpontja nagyon közel esik a legalacsonyabb 
átlaghoz. Ha a kezdıpontot alacsonyabbra vesszük 
(pl. 4,00-ra), akkor sokkal reálisabb a diagram: 
 

4. 

 
 

 

Több hibát is felfedezhetünk. Kördiagramnál 
szerencsésebb a jelmagyarázat helyett a körcikkek 
mellett feliratokat szerepeltetni, mivel így 
könnyebben azonosíthatók az egyes körcikkek. 
Másrészt az adatsor nagyon sok adatpontot 
tartalmaz, a kördiagram áttekinthetetlenné vált. Más 
típussal (pl. oszlopdiagram) sokkal szemléletesebb a 
kép. 

5. 

 
 
 
 

 

Az adatokat nézve látható, hogy az ábrázolt két 
adatsorban az értékek nagyságrendje eltérı (az 
igazolt hiányzásoknál 67-107 közötti, az 
igazolatlanoknál 0 és 12 közötti értékek 
szerepelnek). Ezeket az adatokat egy y-tengelyre 
vetítve a kisebb adatok észrevehetetlenek. Más 
diagramtípust kell használnunk. 
A megoldás, ha kombinált diagramtípussal 
dolgozunk, azaz olyannal, amely két értéktengelyt 
használ. Ilyen pl. a vonal-oszlop 2 tengellyel, vagy a 
vonal, 2 tengellyel típus. Ekkor a két értéktengely (y-
tengely) skálája nagyságrendileg eltérı is lehet. 
 

 

6. 

 
 

 

Esztétikai hibája csak egy van a diagramnak: a 
jelmagyarázat túl sok helyet foglal el a rajzterületbıl. 
Ha a diagram alá helyezzük, sokkal nagyobb helye 
van az oszlopoknak. 
Van azonban egy ennél komolyabb, tartalmi hibája is 
a diagramnak! 
Az elızı feladatban a két adatsor értékei azonos 
jellegőek voltak (mindkettı hiányzás-órát jelentett, 
csak a nagyságrendben volt eltérés). Ebben a 
részfeladatban azonban a két ábrázolandó adatsor 
értékei más jellegőek (az egyik tanulmányi átlag, a 
másik hiányzás-óra). Az adatok nagyságrendje nem 
jelentene problémát, de nincs értelme egy tengelyre 
vetíteni a két, jellegében teljesen eltérı adatsort! 
Ebben az esetben is használjunk két értéktengelyes 
kombinált diagramot. Hogy az értékek 
különbözıségét még jobban hangsúlyozzuk, célszerő 
olyat választani, ahol a két adatsort más jellegő 
minta jelzi (pl. vonal-oszlop 2 tengely). 

Piac/2 

1.  Az állomány megnyitásakor konvertálási varázslót alkalmazunk. Mentés a használt programnak 
megfelelıen. 

2.  F2: =KEREK(D2*C2*E2;0) 

Ha függvény helyett formátumbeállítással dolgozunk, akkor az értéket továbbra is 
tizedesjegyekkel együtt tárolná a gép, és a késıbbi feladatokban hibás eredményeket kapnánk. 
A C és az F oszlopban pénznem formátumot állítunk. 
A rendezéshez kijelöljük az egész adattáblát (célszerő a rovatfejeket is belevenni). Elsıdleges 
rendezési szempont: Napi ár (csökkenı), másodlagos szempont: Kereskedı (emelkedı). 

3.  Bekapcsoljuk az AutoSzőrıt, majd sorban végrehajtjuk a szőrési mőveleteket. Az egyes szőrések 
után mindig jelenítsük meg az összes adatot! 
A feladat végrehajtása után kapcsoljuk ki az AutoSzőrıt! 

4.  Létrehozunk egy új munkalapot, átnevezzük. 
Az Irányított szőrı beállításainál a listatartomány minden esetben az eredeti munkalap A1:F116-
os tartománya. Az adatok más helyre kerüljenek, az új hely kezdıpontját értelemszerően 
határozzuk meg. Az alábbiakban a szőrıtartományokat adjuk meg. 

5.  Ha a szőrt adatokból csak az elsı vagy utolsó valahány tételt szeretnénk megjeleníteni, akkor 
AutoSzőrıvel kell dolgoznunk, tehát az 1. és a 3. feladatban ez a módszer lesz célravezetı. 
A 2. pont feladatának szőrıfeltételét azonban nem tudjuk AutoSzőrıvel megadni, ennél 
Irányított szőrıt használunk: 

 
 
 
 
 
 
 
 
 
 

Naptár 

1.  Az állomány megnyitásakor konvertálási varázslót alkalmazunk. Mentés a használt programnak 
megfelelıen. 

2.  Létrehozunk egy új munkalapot, átnevezzük. Ezen fogjuk elkészíteni a naptárt. 

3.  Beírjuk a szövegeket az A1, B1, C1, D1 cellákba. 
Az A2 ill. B2 cellákba beírjuk a kívánt évszámot illetve hónapszámot. 

4.  C2: =HÉT.NAPJA(DÁTUM(A2;B2;1);2) 
D2: 
=HA(MARADÉK(A2;400)=0;"Igen"; HA(MARADÉK(A2;100)=0; "Nem"; 
HA(MARADÉK(A2;4)=0;"Igen";"Nem"))) 
vagy 
D2: 
=HA(VAGY(ÉS((MARADÉK(A2;4)=0);(MARADÉK(A2;100)<>0)) ;ÉS((MARADÉK(A2;100)=0);(MARADÉK(A2
;400)=0)));"Igen";"Nem")  

5.  A sorozat elsı két tagját beírva kitöltést alkalmazhatunk. 

6.  A4: =ÖSSZEF ŐZ(A2;". ";FKERES(B2;Adatok!E1:F12;2;IGAZ)) 

A cellék egyesítésével elérjük, hogy a címsor a naptár fölött középen helyezkedik el. 

7.  Ha a táblázatkezelı program lehetıvé teszi, használjunk automatikus kitöltést! Ha nem, írjuk be 
a napok neveit! 

8.  A6: =HA(A3<C2;"";1) 
B6: =HA(B$3<$C$2;"";HA(B$3=$C$2;1;A6+1)) 

Ez a képlet átmásolható a C6:G6 tartomány celláiba is. 
A7: 
=HA(A6="";"";HA($D$2="Igen";FKERES(DÁTUM(2004;$B$2; A6);Adatok!$A$1:$B$366;2;IGAZ);FKER
ES(DÁTUM(2003;$B$2;A6);Adatok!$C$1:$D$365;2;IGAZ)))  

A képlet másolható az A7:G7 tartományban, valamint az összes sorban, melyben névnapok 
szerepelnek (7., 9., 11., 13., 15., 17. sorok A-G oszlopokba esı cellái.) 
A 8., 10., 12. sorokba garantáltan teljes hét kerül, így itt a képlet egyszerősödik: 
A8: =G6+1 
A10: =G8+1 
A12: =G10+1 
B8: ==A8+1 

Ez utóbbi képlet másolható a B8:G8, B10:G10, B12, G12 tartományban. 
A 14., illetve 16. sorok nem feltétlenül adnak teljes hetet, így itt a képlet: 
A14: 
=HA($D$2="Igen";HA(FKERES($B$2;Adatok!$E$1:$H$12;4; IGAZ)>Naptár!G12;Naptár!G12+1;"");H
A(FKERES($B$2;Adatok!$E$1:$H$12;3;IGAZ)>Naptár!G12; Naptár!G12+1;"")) 

Ez másolható az A16 cellába is. 
B14: 
=HA($D$2="Igen";HA(FKERES($B$2;Adatok!$E$1:$H$12;4; IGAZ)>Naptár!A14;Naptár!A14+1;"");H
A(FKERES($B$2;Adatok!$E$1:$H$12;3;IGAZ)>Naptár!A14; Naptár!A14+1;""))  

Ez a képlet másolható a B14:G14, B16:G16 tartományokban. 

9.  Elvégezzük a megadott cellaformázásokat (Formátum/Cellák… menüpont). 
 

Pascal-háromszög 

1.  A forrásállomány importálása után a mentéshez pontos állománynév és formátumállítás 
szükséges. 

2.  A munkalapot nevezzük el „Pascal-hárömszög” néven. 

3.  Az oszlopszélességet és a sormagasságot a megadott értékre állítsuk be, mert így lesz látványos a 
kialakuló minta. 

4.  A beírt szám lesz az osztó, a késıbbiek során erre kell hivatkozni. Értékét a késıbbiekben 
érdemes változtatni, ezáltal más-más mintát kapunk. 

5.  A1: =HA(Alapadatok!A1<>0;MARADÉK(Alapadatok!A1;$BK$ 33);"") 

A képlet kialakítása során figyeljünk a megfelelı feltételre és az osztóra való abszolút 
hivatkozásra. 

6.  A másolás elıtt ellenırizzük, hogy a képletünk helyesen hivatkozik-e a megadott cellákra 
(abszolút hivatkozás szükséges)! 

7.  A feltételes formázás feltételénél az egyenlı-t és a nulla értéket állítsuk be. 
A feltételes formázásnál ne csak a háttér színét állítsuk be a megadott színre, hanem a betők 
színét is! 

Darázs 

1.  A forrásállomány importálása után a mentéshez pontos állománynév és formátumállítás 
szükséges. A cellákban F, S és V karakterek jelennek meg, illetve üresen maradnak a cellák. 

2.  Az oszlopszélességet és a sormagasságot a megadott értékre állítsuk be ahhoz, hogy a grafika 
elıtőnjön. 

3.  A feladat lényeges része a feltételes formázás használata. A cellák megadott karakterekkel való 
egyenlıségét vizsgáljuk. Megfelelı háttér- és írásszínt válasszunk! 

 
4.  AI34: =DARABTELI($A$1:$AG$32;AH34) 

Az elsı paraméter (a vizsgált tartomány) hivatkozásának rögzítése a függvény másolása miatt 
lényeges. A második paraméter relatív hivatkozása biztosítja a megfelelı karakterek számának 
meghatározását. Eredmény a forrásadatok alapján: 

Színek Száma 
F 228 
S 100 
V 133  

5.  A megoldás összetett, emiatt nehezebb. A függvénykonstrukció módszere gyakorolható vagy 
mutatható be ezzel. 
Munkalapok közötti hivatkozást használunk. Lássuk be, hogy az eredeti kép minden pontjának 
(cellájának) 4 pont (cella) felel meg kétszeres nagyításnál. 
A következı függvényt másoljuk az új munkalapon az eredeti kép vízszintes és függıleges 
méretének kétszeresére. 
=INDEX(darazs!$A$1:$AG$32;INT(SOR(darazs!A1)/2)+1;I NT(OSZLOP(darazs!A1)/2)+1) 

Az új kép táblázatát az eredeti képhez hasonlóan formázzuk. (A nullák zavaró megjelenését fehér 
írásszínnel küszöbölhetjük ki.) 
A feltételes formázást a formátum másolásával kényelmesen és gyorsan végezhetjük el. 

 


